

EXEMPLES CAST3M

F. DI PAOLA

EXEMPLES D'UTILISATION DE LA
PROCÉDURE PASAPAS



ÉDITION 2011

Documentation Cast3M 2011

<http://www-cast3m.cea.fr>

Cast3M est un logiciel de calcul par la méthode des éléments finis pour la mécanique des structures et des fluides. Cast3M est développé au Département de Modélisation des Systèmes et Structures (DM2S) de la Direction de l'Énergie Nucléaire du Commissariat à l'Énergie Atomique et aux Énergies Alternatives (CEA).

Le développement de Cast3M entre dans le cadre d'une activité de recherche dans le domaine de la mécanique dont le but est de définir un instrument de haut niveau, pouvant servir de support pour la conception, le dimensionnement et l'analyse de structures et de composants.

Dans cette optique, Cast3M intègre non seulement les processus de résolution (solveur) mais également les fonctions de construction du modèle (pré-processeur) et d'exploitation des résultats (post-traitement). Cast3M est un logiciel « boîte à outils » qui permet à l'utilisateur de développer des fonctions répondant à ses propres besoins.

Cast3M est notamment utilisé dans le secteur de l'énergie nucléaire, comme outil de simulation ou comme plateforme de développement d'applications spécialisées. En particulier, Cast3M est utilisé par l'Institut de Radioprotection et de Sécurité Nucléaire (IRSN) dans le cadre des analyses de sûreté des installations nucléaires françaises.



Table des matières

| | | |
|----------|---|-----------|
| 1 | Thermique transitoire | 7 |
| 1.1 | Description | 7 |
| 1.2 | Objectifs | 7 |
| 1.3 | Conditions initiales et aux limites | 8 |
| 1.4 | Propriétés thermodynamiques | 8 |
| 1.5 | Commentaires | 8 |
| 1.5.1 | Méthode de résolution | 8 |
| 1.5.2 | Paramètres matériaux dépendant de la température | 8 |
| 1.5.3 | Prise en compte du changement de phase | 9 |
| 1.5.4 | Prise en compte des conditions aux limites de rayonnement | 9 |
| 1.5.5 | Source dépendante de la température (CHARTHER) | 10 |
| 1.6 | Résultats numériques | 11 |
| 1.7 | Jeu de données | 13 |
| 2 | Thermo-mécanique | 15 |
| 2.1 | Description | 15 |
| 2.2 | Objectifs | 15 |
| 2.3 | Conditions initiales et aux limites | 15 |
| 2.4 | Propriétés thermo-mécaniques | 16 |
| 2.5 | Commentaires | 16 |
| 2.5.1 | Prise en compte du contact | 16 |
| 2.5.2 | Méthode de résolution | 16 |
| 2.5.3 | Modification des conditions aux limites (PERSO1) | 16 |
| 2.6 | Modification des conditions aux limites (REEV_MEC) | 17 |
| 2.7 | Paramètres matériaux dépendant de l'état mécanique (REEV_MEC) | 18 |
| 2.8 | Résultats numériques | 19 |
| 2.9 | Jeu de données | 22 |
| 3 | Mécanique | 25 |
| 3.1 | Description | 25 |
| 3.2 | Objectifs | 25 |
| 3.3 | Conditions initiales et aux limites | 25 |
| 3.4 | Propriétés mécaniques | 26 |
| 3.5 | Commentaires | 26 |
| 3.5.1 | Prise en compte du contact | 26 |
| 3.5.2 | Calcul de la longueur caractéristique | 26 |
| 3.5.3 | Modélisation de la fissuration en traction | 27 |
| 3.5.4 | Élimination des éléments au cours du calcul (PERSO1) | 27 |
| 3.6 | Résultats numériques | 28 |
| 3.7 | Jeu de données | 30 |



Chapitre 1

Cas test 1 : Thermique transitoire non linéaire dans une plaque

1.1 Description

La figure 1.1 présente le problème de ce cas test. Une plaque plane est chauffée par une source de chaleur située sur l'un de ses coins. Elle est refroidie par convection sur deux de ses bords et par rayonnement sur l'un de ses bords. On modélise également le changement de phase du matériau. Enfin, on choisit de faire dépendre certains paramètres du problème de la température :

- la conductivité thermique ;
- la masse volumique ;
- le coefficient d'échange convectif ;
- l'intensité de la source de chaleur.




FIG. 1.1 – Illustration du problème.

1.2 Objectifs

- Tester la gestion de la thermique transitoire avec plusieurs modes de transfert de chaleur (conduction, convection, rayonnement), une source de chaleur et du changement de phase.
- Faire dépendre les paramètres du problème de la température (paramètres matériaux et conditions aux limites).

- Montrer l'utilisation de la procédure utilisateur CHARTHER dans PASAPAS.

1.3 Conditions initiales et aux limites

- Température initiale dans la plaque : 20 °C.
- Source de chaleur d'intensité variable de 3 à 5 MW.m⁻³ selon la température.
- Convection sur LHAUT et LDROITE : $T_{ec} = 20$ °C et h variable de 20 à 300 W.m⁻².K⁻¹ selon la température.
- Rayonnement à l'infini sur LDROITE : $T_{er} = 20$ °C et $\varepsilon = 0.2$.

1.4 Propriétés thermodynamiques

Les propriétés thermodynamiques de la plaque sont présentées dans le tableau 1.1. Elles correspondent au cuivre. Certaines propriétés dépendent de la température.

TAB. 1.1 – Propriétés thermodynamiques.

| Propriété | Température (°C) | Valeur |
|--|------------------|-----------------------|
| Masse volumique (kg.m ⁻³) | 20 | 8930 |
| | 1083 | 7940 |
| | 1100 | 7924 |
| | 1200 | 7846 |
| | 1300 | 7764 |
| Conductivité thermique (W.m ⁻¹ .K ⁻¹) | 0 | 401 |
| | 27 | 398 |
| | 727 | 357 |
| Capacité calorifique massique (J.kg ⁻¹ .K ⁻¹) | — | 385 |
| Température de fusion (°C) | — | 1083 |
| Enthalpie massique de fusion (J.kg ⁻¹) | — | 204.8 10 ³ |

1.5 Commentaires

1.5.1 Méthode de résolution

L'algorithme choisi pour la résolution de la thermique est celui de la theta-méthode avec un coefficient de relaxation θ égal à 1, ce qui correspond à un schéma temporel implicite pur. Ceci est demandé par les commandes suivantes :

```
TAB1 . 'PROCEDURE_THERMIQUE'      = 'NONLINEAIRE' ;
TAB1 . 'RELAXATION_THETA'         = 1. ;
```

1.5.2 Paramètres matériaux dépendant de la température

La prise en compte de paramètres matériaux dépendants de la température est simple à réaliser. Il suffit, pour cela de déclarer, dans l'opérateur MATE qui crée le champ de propriété matérielles, des objets de type EVOLUTION pour les composantes désirées. Ces évolutions doivent avoir comme nom d'abscisse 'T' et comme nom d'ordonnée la composante utilisée par MATE. Ainsi, dans notre exemple, la conductivité thermique (composante 'K'), la masse volumique ('RHO') et le coefficient d'échange convectif ('H') sont données par des



1.5. COMMENTAIRES

évolutions, alors que la capacité thermique ('C'), la température de fusion ('TPHA') l'enthalpie massique de fusion ('QLAT') et l'émissivité ('EMIS') sont des flottants :

```
** Materiaux
LAMBDA  = EVOL 'MANU' 'T' (PROG 0. 27. 727.)
          'K' (PROG 401. 398. 357.) ;
CP       = 385. ;
RHO      = EVOL 'MANU' 'T' (PROG 20. 1083. 1100. 1200. 1300.)
          'RHO' (PROG 8930. 7940. 7924. 7846. 7764.) ;
TFUS     = 1083. ;
QLATFUS  = 204.8E3 ;
** Conditions aux limites
EMISCU   = 0.2 ;
HCONV    = EVOL 'MANU' 'T' (PROG 20. 700. 1100. 1500.)
          'H' (PROG 20. 100. 200. 300.) ;
```

Les modèles de conduction, convection et rayonnement, ainsi que les champs de propriétés matérielles associés à ces modèles, sont simplement déclarés :

```
MOD_THER = MODE MAILT 'THERMIQUE' 'PHASE' ;
MAT_THER = MATE MOD_THER 'K' LAMBDA 'C' CP 'RHO' RHO
          'TPHA' TFUS 'QLAT' QLATFUS ;
MOD1      = REDU MOD_THER MAIL1 ;
MOD_CONV  = MODE LEXT 'CONVECTION' ;
MAT_CONV  = MATE MOD_CONV 'H' HCONV ;
MOD_RAY   = MODE LDROITE 'RAYONNEMENT' ;
MAT_RAY   = MATE MOD_RAY 'EMIS' EMISCU ;
```

Le modèle MOD1 est la réduction du modèle de conduction au maillage MAIL1 et sera utilisé pour appliquer la source de chaleur. Notons que l'on peut faire dépendre les paramètres matériaux d'autre variables que la température. Pour cela, il faut utiliser des évolutions ou le nom d'abscisse est un mot clé choisi. Il faudra alors utiliser un objet de type CHARGEMENT, construit avec ce mot clé, pour donner l'évolution du paramètre matériau en fonction de la variable (cf. le cas test 2 à la section 2.7).

1.5.3 Prise en compte du changement de phase

La prise en compte du changement de phase est simple. Il suffit d'ajouter le mot clé 'PHASE' au modèle de conduction du matériau, puis de préciser la température et l'enthalpie massique de changement d'état dans le champ de propriétés matérielles, en plus des paramètres « classiques » :

```
MOD_THER = MODE MAILT 'THERMIQUE' 'PHASE' ;
MAT_THER = MATE MOD_THER 'K' LAMBDA 'C' CP 'RHO' RHO
          'TPHA' TFUS 'QLAT' QLATFUS ;
```

1.5.4 Prise en compte des conditions aux limites de rayonnement

Les conditions aux limites de rayonnement requièrent un traitement particulier. Après avoir défini un modèle de rayonnement sur la ligne LDROITE et un champ de propriétés matérielles associé (émissivité), on crée un chargement, avec le mot clé 'TERA', qui définit l'évolution de la température extérieure du milieu sur le maillage et en fonction du temps :

```
CHTRAY   = MANU 'CHPO' LDROITE 1 'T' T_EXT ;
CHA2     = CHAR 'TERA' CHTRAY EV1 ;
```

Ici, la température est constante dans l'espace et dans le temps. Pour la faire varier dans l'espace, on définira un champ par point (CHTRAY) non uniforme sur le maillage. Pour la faire varier dans le temps, on précisera une évolution non uniforme (EV1).

Lors de la création de la table pour PASAPAS, il ne faut pas intégrer le modèle de rayonnement à l'indice 'MODELE' de la table. En revanche, il faut intégrer le champ de propriétés matérielles et le chargement de rayonnement aux indices 'CARACTERISTIQUES' et 'CHARGEMENT' :

```
TABL                                = TABL ;
TABL . 'MODELE'                      = MOD_THER ET MOD_CONV ;
TABL . 'CARACTERISTIQUES'           = MAT_THER ET MAT_CONV ET MAT_RAY ;
TABL . 'CHARGEMENT'                 = CHA1 ET CHA2 ;
```

Le modèle de rayonnement est précisé dans une sous-table d'indice 'RAYONNEMENT' qui contient autant de sous-sous-tables que de zones concernées par le rayonnement (ici une seule). Ces sous-sous-tables doivent être indicées par des entiers consécutifs (1, 2, ...). Plusieurs informations sont à renseigner selon le type de rayonnement. Dans le cas d'un rayonnement à l'infini on précise le modèle¹ :

```
TABL . 'RAYONNEMENT'                = TABL ;
TABL . 'RAYONNEMENT' . 1            = TABL ;
TABL . 'RAYONNEMENT' . 1 . 'TYPE'   = 'INFINI' ;
TABL . 'RAYONNEMENT' . 1 . 'MODELE' = MOD_RAY ;
```

Il est indispensable de préciser si l'on travaille avec des degrés CELSIUS car, dans le cas du rayonnement, les relations flux/température sont non linéaires. C'est le cas ici :

```
TABL . 'CELSIUS'                    = VRAI ;
```

1.5.5 Source dépendante de la température (CHARTHER)

La prise en compte d'une source de chaleur se fait classiquement en créant :

- un champ de flux de chaleur nodal (second membre) correspondant à la source à l'aide de l'opérateur SOUR ;
- puis un chargement, avec le mot clé 'Q', précisant l'évolution de ce champ en fonction du temps.

Dans notre problème, on souhaite faire dépendre l'intensité de la source de chaleur de la température. Pour cela, il faut re-calculer le second membre du problème à chaque itération de la boucle de convergence selon la valeur de la température obtenue à l'itération précédente. On utilise, pour cela, la procédure CHARTHER qui est appelée avant l'entrée dans la boucle de convergence puis après chacune de ses itérations. Elle permet d'ajouter un terme au second membre (ainsi qu'au premier). Une version de cette procédure est déjà fournie dans CAST3M et traite la prise en compte du rayonnement dans PASAPAS. Nous allons donc ajouter des instructions à cette procédure pour réaliser ce que l'on désire. Pour ne pas perdre les fonctionnalités originelles de CHARTHER, nous l'avons recopiée dans le jeu de données.

Avant toute chose, nous créons, au début du fichier, une évolution reliant l'intensité de la source de chaleur à la température :

```
SOURVST = EVOL 'MANU' 'T' (PROG 20. 150. 300. 500. 700. 1000. 1500.)
          'Q' (1.E6*(PROG 3.5 3. 2.4 2.1 2. 1.9 1.8)) ;
```

L'utilisation de la procédure utilisateur CHARTHER est demandée par l'instruction :

```
TABL . 'PROCEDURE_CHARTHER'         = VRAI ;
```

¹On peut aussi préciser une valeur d'émissivité à l'infini, qui est prise égale à 1 par défaut.



1.6. RÉSULTATS NUMÉRIQUES

De base, la procédure admet comme argument d'entrée la table de PASAPAS, nommée PRECED dans la procédure. En sortie, elle fournit une table, nommée ici TAA, contenant le premier et le second membre à ajouter au problème pour modéliser le rayonnement. Dans le cas de la source de chaleur dépendante de la température, nous ajoutons les lignes suivantes à la fin de la procédure :

```
** COMPLEMENT POUR LA PROCEDURE CHARTHER
CHPO_T  = PRECED . 'ESTIMATION' . 'TEMPERATURES' ;
TMAIL1  = REDU CHPO_T MAIL1 ;
CHPOSOUR = IPOL TMAIL1 SOURVST ;
CHSOUR  = SOUR MOD1 CHPOSOUR ;
TAA . 'ADDI_SECOND' = (TAA . 'ADDI_SECOND') ET CHSOUR ;
FINP TAA;
```

Commentons ligne par ligne ces instructions :

- On récupère d'abord le champ de température calculé à la dernière itération. Ce dernier est contenu dans l'indice 'ESTIMATION' de la table de PASAPAS.
- On réduit alors ce champ au maillage MAIL1 sur lequel s'applique la source de chaleur.
- On calcule le champ de source de chaleur à partir du champ de température et de l'évolution SOURVST par interpolation.
- On calcule le champ de flux nodaux (second membre) correspondant à la source de chaleur.
- On ajoute ce second membre à l'indice 'ADDI_SECOND' déjà existant de la table TAA.
- Cette table est donnée en sortie de la procédure.

La procédure de résolution thermique (TRANSNON dans notre cas) se chargera alors d'ajouter les nouveaux termes de premier et de second membre pour la résolution du problème.

1.6 Résultats numériques

On observe l'évolution de la température le long de la ligne inférieure du maillage LBAS. La figure 1.2 présente l'évolution de ce profil de température pour différents instants de calcul.




FIG. 1.2 – Évolution de la température le long de la ligne LBAS pour différents instants de calcul. T_f correspond au temps final de calcul.



1.7 Jeu de données

```
OPTI 'ECHO' 0 ;
*****
*
*      Chauffage d'une plaque par une source de chaleur
*
* Certains parametres dependent de la temperature:
* - la conductivite
* - la masse volumique
* - le coefficient d'echange convectif
* - l'intensite de la source de chaleur
*
*****
*
* Calcul thermique transitoire.
* Utilisation de la procedure PASAPAS.
* Utilisation de la procedure utilisateur CHARITHER.
* Transfert par conduction, convection et rayonnement
* Changement de phase
*
*
*      Convection (h = 20 - 300 W.m-2.K-1)
*      (Text = 20 degres C )
*
*
*      T initiale = 20 degres C
*
*      Convection
*      (h = 20 - 300 W.m-2.K-1)
*      (Text = 20 degres C )
*      +
*      Rayonnement
*      (e = 0.2 )
*      (Text = 20 degres C)
*
*
*      Source de
*      chaleur
*      3 - 5
*      MW.m-3
*
*
* <---- R ---->
*
* <----- L ----->
*
*
* On etudie la diffusion thermique dans la plaque au cours du temps
*
* Caracteristiques de la plaque (cuivre) :
*
* L = 1. m
* R = 0.25 m
* K = 357 - 401 W.m-1.K-1
* Cp = 385 J.kg-1.K-1
* Rho = 7764 - 8930 kg.m-3
* T_fus = 1083 degres C
* Q_lat = 204.8E3 J.kg-1
* Emis = 0.2
*
*----- Options generales -----*
*
OPTI 'DIME' 2 'ELEM' 'QUA4' 'MODE' 'PLAN' 'DEFO' ;
** Indicateur de trace
GRAPH = VRAI ;
*
*----- Parametres geometriques, materiaux et CL -----*
*
L = 1. ;
R = 0.25 ;
** Materiaux
LAMBDA = EVOL 'MANU' 'T' (PROG 0. 27. 727.)
          'K' (PROG 401. 398. 357.) ;
CP = 385. ;
RHO = EVOL 'MANU' 'T' (PROG 20. 1083. 1100. 1200. 1300.)
          'RHO' (PROG 8930. 7940. 7924. 7846. 7764.) ;
TFUS = 1083. ;
QLATFUS = 204.8E3 ;
EMISCU = 0.2 ;
** Conditions aux limites
HCONV = EVOL 'MANU' 'T' (PROG 20. 700. 1100. 1500.)
          'H' (PROG 20. 100. 200. 300.) ;
SOURVST = EVOL 'MANU' 'T' (PROG 20. 150. 300. 500. 700. 1000. 1500.)
          'Q' (1.E6*(PROG 5. 4. 3.7 3.5 3.3 3.1 3.)) ;
T_INI = 20. ;
T_EXT = 20. ;
** Temps de calcul final
TEMPS_F = 1.E5 ;
*
*----- Maillage -----*
*
OPTI 'DENS' (R/5.) ;
P0 = 0. 0. ;
P1 = L 0. ;

LBAS = DROI P0 P1 ;
S1 = (LBAS TRAN (0. L)) COUL 'BLEU' ;
CHX = S1 COOR 1 ;
PTX = CHX POIN 'EGINFE' (R + 1.E-8) ;
EL1 = S1 ELEM 'APPU' 'STRI' PTX ;
CHY = EL1 COOR 2 ;
PTY = CHY POIN 'EGINFE' (R + 1.E-8) ;
MAIL1 = (EL1 ELEM 'APPU' 'STRI' PTY) COUL 'ROUG' ;
MAIL2 = DIFF S1 MAIL1 ;
MAILT = MAIL1 ET MAIL2 ;
LDROITE = S1 COTE 2 ;
LHAUT = S1 COTE 3 ;
LEXT = LDROITE ET LHAUT ;
SI GRAPH ;
TRAC 'FACE' MAILT 'TITR' 'Maillage' ;
FINS ;
*
*----- Modeles et materiaux -----*
*
MOD_THER = MODE MAILT 'THERMIQUE' 'PHASE' ;
MAT_THER = MATE MOD_THER 'K' LAMBDA 'C' CP 'RHO' RHO
          'TPHA' TFUS 'QLAT' QLATFUS ;
MOD1 = REDU MOD_THER MAIL1 ;
MOD_CONV = MODE LEXT 'CONVECTION' ;
MAT_CONV = MATE MOD_CONV 'H' HCONV ;
MOD_RAY = MODE LDROITE 'RAYONNEMENT' ;
MAT_RAY = MATE MOD_RAY 'EMIS' EMISCU ;
*
*----- Chargements thermiques -----*
*
EV1 = EVOL 'MANU' (PROG 0. TEMPS_F) (PROG 1. 1.) ;
** Convection
CHTCONV = MANU 'CHPO' LEXT 1 'T' T_EXT ;
CHA1 = CHAR 'TECO' CHTCONV EV1 ;
** Rayonnement
CHTRAY = MANU 'CHPO' LDROITE 1 'T' T_EXT ;
CHA2 = CHAR 'TERA' CHTRAY EV1 ;
*
*----- Champ de temperature initiale -----*
*
TEM0 = MANU 'CHPO' MAILT 1 'T' T_INI ;
*
*----- Creation de la table pour PASAPAS -----*
*
LTSPCAL1 = PROG 0. PAS (TEMPS_F/20.) TEMPS_F ;
TAB1 = TABL ;
TAB1 . 'MODELE' = MOD_THER ET MOD_CONV ;
TAB1 . 'CARACTERISTIQUES' = MAT_THER ET MAT_CONV ET MAT_RAY ;
TAB1 . 'TEMPERATURES' = TABL ;
TAB1 . 'TEMPERATURES' . 0 = TEM0 ;
TAB1 . 'CHARGEMENT' = CHA1 ET CHA2 ;
TAB1 . 'PROCEDURE_THERMIQUE' = 'NONLINEAIRE' ;
TAB1 . 'RELAXATION_THETA' = 1. ;
TAB1 . 'RAYONNEMENT' = TABL ;
TAB1 . 'RAYONNEMENT' . 1 = TABL ;
TAB1 . 'RAYONNEMENT' . 1 . 'TYPE' = 'INFINI' ;
TAB1 . 'RAYONNEMENT' . 1 . 'MODELE' = MOD_RAY ;
TAB1 . 'TEMPS_CALCULES' = LTSPCAL1 ;
TAB1 . 'CELSIUS' = VRAI ;
TAB1 . 'PROCEDURE_CHARITHER' = VRAI ;
*
*----- Procedure CHARITHER -----*
*
DEBP CHARITHER PRECED*TABLE' TT*'PLOTTANT' ;
*MESS ' Entree dans CHARITHER TT vaut ' TT ' *****' ;
TAA = TABL ;
ETAB = PRECED . 'WTABLE' ;
SI (ETAB . 'RAYO') ;
SI (EXIS ETAB 'CTE_STEFAN_BOLTZMANN') ;
CTE_SB = ETAB . 'CTE_STEFAN_BOLTZMANN' ;
SINO ;
CTE_SB = 5.673E-8 ;
MESS 'ATTENTION ! Constante de Stefan-Boltzmann en unités SI' ;
MESS ' soit ' CTE_SB ;
FINS ;
ICALC = VRAI ;
SI (EGA (ETAB . 'DEJA_CALC_CHARITHER') TT) ;
ICALC = FAUX ;
FINS ;
SI ICALC ;
ETAB . 'DEJA_CALC_CHARITHER' = TT ;
RAYO_VAL = ETAB . 'RAYONNEMENT' ;
NDIM3 = DIME RAYO_VAL ;
REPE BOU_RAY NDIM3 ;
SI (EGA 'CAVITE' (RAYO_VAL . (&BOU_RAY) . 'TYPE')) ;
SI ((ETAB . 'REA_GEOM') OU
(NON (EXIS (RAYO_VAL . &BOU_RAY) 'FAC_FORM')) ;
CAV_CONV = RAYO_VAL . &BOU_RAY . 'CONVEXE' ;
CAV_FERM = RAYO_VAL . &BOU_RAY . 'FERME' ;
SI CAV_CONV ;
SI CAV_FERM ;
RAYO_VAL . &BOU_RAY . 'FAC_FORM' = FFOR
(RAYO_VAL . &BOU_RAY . 'MODELE') 'CVXE' ;
SINO ;
```




Chapitre 2

Variation d'un jeu sous l'action d'une sollicitation thermique en régime transitoire

2.1 Description

La figure 2.1 présente le problème de ce cas test. Deux barreaux de longueur $L = 1$ m et de propriétés thermodynamiques différentes sont séparés par un jeu de 1 cm. Le barreau 1 est chauffé à une extrémité alors que le barreau 2 est maintenu à sa température initiale. Sous l'action de cette sollicitation thermique, le barreau 1 va se dilater et réduire le jeu jusqu'à entrer en contact avec le barreau 2. La mise en contact des deux barreaux permet le transfert de chaleur par conduction entre les barreaux. Le barreau 2 refroidit le barreau 1 ce qui provoque sa rétractation et permet de ré-ouvrir le jeu. La sollicitation thermique, maintenue dans le barreau 1, provoque à nouveau sa dilatation, la réduction du jeu et la mise en contact. Le système est donc instable.




FIG. 2.1 – Illustration du problème.

2.2 Objectifs

- Tester la réduction d'un jeu sous l'action d'une sollicitation thermique.
- Montrer l'utilisation des procédures utilisateurs PERSO1 et REEV_MEC dans PASAPAS.

2.3 Conditions initiales et aux limites

- Température initiale dans les deux barreaux : 20 K.
- Température imposée en x_1 : $T_1 = 500$ K.
- Température imposée en x_4 : $T_4 = 20$ K.
- Encastrement en x_1 .
- Encastrement en x_4 .

– Déplacements supprimés sur l'axe vertical.

2.4 Propriétés thermo-mécaniques

Les propriétés thermo-mécaniques des barreaux sont présentées dans le tableau 2.1.

TAB. 2.1 – Propriétés thermo-mécaniques

| Propriété | Barreau 1 | Barreau 2 |
|--|-----------|-----------|
| Conductivité thermique ($\text{W.m}^{-1}.\text{K}^{-1}$) | 60 | 80 |
| Capacité calorifique massique ($\text{J.kg}^{-1}.\text{K}^{-1}$) | 800 | 100 |
| Masse volumique (kg.m^{-3}) | 7800 | 7800 |
| Dilatation thermique (10^{-6}K^{-1}) | 12 | 12 |
| Module de YOUNG (GPa) | 200 | 200 |
| Coefficient de POISSON | 0.3 | 0.3 |

2.5 Commentaires

2.5.1 Prise en compte du contact

La condition de contact entre les deux extrémités des barreaux (lignes L2 et L3) est imposée au moyen de l'opérateur IMPO :

```
MCONT1 = IMPO 'MAIL' L2 L3 'COUL' 'JAUN' ;
```

Le maillage ainsi créé doit être passé en argument de la table de PASAPAS à l'indice 'CONTACT' :

```
TAB1 . 'CONTACT' = MCONT1 ;
```

2.5.2 Méthode de résolution

L'algorithme choisi pour la résolution de la thermique est celui de la theta-méthode avec un coefficient de relaxation θ égal à 1, ce qui correspond à un schéma temporel implicite pur. Ceci est demandé par les commandes suivantes :

```
TAB1 . 'PROCEDURE_THERMIQUE' = 'NONLINEAIRE' ;
TAB1 . 'RELAXATION_THETA' = 1. ;
```

On demande également à itérer les calculs thermique et mécanique dans la boucle de convergence thermique-mécanique car les résultats des deux calculs sont dépendants l'un de l'autre et on précise la valeur du critère de convergence (égal à 10^{-2} par défaut) :

```
TAB1 . 'CONVERGENCE_MEC_THE' = VRAI ;
TAB1 . 'CRITERE_COHERENCE' = 1.E-3 ;
```

2.5.3 Modification des conditions aux limites (PERSO1)

Le transfert de chaleur entre les deux barreaux se fait par conduction uniquement lorsqu'il y a contact. Ce transfert est modélisé en imposant une relation d'égalité sur les températures des lignes L2 et L3 seulement lors du contact. Il est donc nécessaire de modifier, au cours du calcul, les paramètres du problème thermique selon l'état mécanique du système. Ceci peut être fait au moyen de la procédure utilisateur PERSO1. Cette procédure est appelée par PASAPAS après la résolution de chaque pas de calcul. Son utilisation est demandée par la commande :



2.6. MODIFICATION DES CONDITIONS AUX LIMITES (REEV_MEC)

```
TAB1 . 'PROCEDURE_PERSON1' = VRAI ;
```

On prend soin de stocker les blocages thermiques initiaux dans la table de PASAPAS. On crée également un indice indiquant s'il y a contact ou non :

```
TAB1 . 'ICONCONTACT' = FAUX ;  
TAB1 . 'MES_DONNEES' = TABL ;  
TAB1 . 'MES_DONNEES' . 'BLOCAGES_THERMIQUES' = CL_TH ;
```

La procédure PERSON1 est ensuite définie pour modifier, comme on le souhaite, la table de PASAPAS. On commence par récupérer le champ de déplacements calculé au dernier pas (contenu dans l'indice 'ESTIMATION') ainsi que la table de travail ('WTABLE') qui contient, entre autres, les blocages thermiques que l'on va modifier :

```
DEP = TTU . 'ESTIMATION' . 'DEPLACEMENTS' ;  
WTAB = TTU . 'WTABLE' ;
```

Le champ de déplacements permet de calculer la valeur du jeu au dernier pas de temps. Si ce jeu est fermé, les barreaux sont considérés comme en contact. La condition d'égalité des températures est créée avec l'opérateur RELA :

```
CL6 = RELA 'T' L2 - 'T' L3 ;
```

Cette condition est ajoutée à l'indice 'BLOCAGES_THERMIQUES' de la table de travail et l'on actualise également l'indicateur de contact :

```
TTU . 'ICONCONTACT' = VRAI ;  
WTAB . 'BLOCAGES_THERMIQUES' =  
(TTU . 'MES_DONNEES' . 'BLOCAGES_THERMIQUES') ET CL6 ;
```

Si le jeu est ouvert, on revient aux blocages thermiques initiaux et on met également à jour l'indicateur de contact :

```
TTU . 'ICONCONTACT' = FAUX ;  
WTAB . 'BLOCAGES_THERMIQUES' =  
(TTU . 'MES_DONNEES' . 'BLOCAGES_THERMIQUES') ;
```

2.6 Modification des conditions aux limites (REEV_MEC)

Les modifications demandées dans la procédure PERSON1 sont effectuées après chaque pas de temps. Ainsi, si à la fin d'un pas de temps, le jeu se referme, la modification des blocages thermiques n'est effective qu'au pas suivant, ce qui n'est pas réaliste. Ce problème peut être résolu en utilisant la procédure REEV_MEC à la place de la procédure PERSON1, pour faire les mêmes modifications. En effet, cette procédure est appelée après la résolution de la mécanique, dans la boucle de convergence thermique-mécanique. Elle permet de réévaluer l'état du système en fonction de l'état mécanique tout juste calculé. Il faut donc déclarer, en entrée de PASAPAS, que l'on souhaite utiliser la procédure utilisateur REEV_MEC au lieu de PERSON1 :

```
TAB1 . 'PROCEDURE_REAC_MECA' = VRAI ;
```

Il suffit alors de renommer la procédure PERSON1 en REEV_MEC dans le jeu de données existant :

```
DEBP REEV_MEC TTU*'TABLE' ;
```

Le contenu de la procédure REEV_MEC peut rester identique à celle utilisée précédemment. La différence entre les deux approches est visible sur les résultats numériques (cf. 2.8).

2.7 Paramètres matériaux dépendant de l'état mécanique (REEV_MEC)

Nous montrons ici les instructions Gibiane à utiliser pour faire dépendre un paramètre matériau d'une variable relative à l'état mécanique du système. On propose, par exemple, de modéliser le transfert de chaleur entre les deux barreaux, lorsqu'il y a contact, non plus par conduction, mais par convection avec un coefficient d'échange dépendant de la pression de contact (ceci permet de représenter le fait qu'un jeu fermé ne conduit pas la chaleur aussi bien que dans le matériau plein). Pour cela, il faut créer un élément de raccord entre les deux lignes du jeu :

```
MRACC = RACC (1.1*JEU_INI) L2 L3 ;
```

Il faut alors créer un modèle de convection sur cet élément de raccord :

```
MODRACC = MODE MRACC 'CONVECTION' ;
```

et créer un champ de propriété matériau où le coefficient d'échange est une évolution qui dépend d'un paramètre, que l'on nomme ici PC :

```
HVSP = EVOL 'MANU' 'PC' (PROG 0. 5.E8 5.1E8) 'H' (PROG 0. 1.E4 1.E4) ;
MATRACC = MATE MODRACC 'H' HVSP ;
```

Ce paramètre PC représente la pression de contact à l'interface. Lorsque la pression est nulle, le coefficient d'échange est nul. Lorsque la pression est positive, le coefficient d'échange augmente, jusqu'à une valeur seuil de pression au delà de laquelle le coefficient d'échange est constant. Ce modèle et ce champ matériau sont, bien entendu, à intégrer aux indices 'MODELE' et 'CARACTERISTIQUES' de la table de PASAPAS. Il faut alors créer un objet chargement avec le mot clé 'PC' sur l'élément de raccord :

```
CH_PC = MANU 'CHPO' MRACC 'PC' 0. ;
CHA2 = CHAR 'PC' CH_PC EV1 ;
```

Ce chargement a comme paramètre PC et est utilisé par CAST3M pour calculer la valeur du coefficient d'échange. La valeur initiale de PC est ici nulle car il n'y a pas de contact. Ce chargement est à intégrer à l'indice 'CHARGEMENT' de la table de PASAPAS. Il s'agit alors de modifier ce chargement au cours du calcul pour actualiser la valeur de PC. Ceci est fait via la procédure utilisateur REEV_MEC. On prend soin de stocker les valeurs initiales des chargements mécaniques et thermiques en entrée de la table de PASAPAS :

```
TAB1 . 'MES_DONNEES' = TABL ;
TAB1 . 'MES_DONNEES' . 'CHAR_MECA' = CHA1 ;
TAB1 . 'MES_DONNEES' . 'CHAR_THER' = CHA2 ;
```

Dans la procédure REEV_MEC, on calcule la valeur de la pression de contact au point P3 à partir des réactions contenues dans la table 'ESTIMATION', puis on modifie le chargement dans la table de travail :

```
DEBP REEV_MEC TTU*'TABLE' ;
** Déplacements, réactions, et table de travail a la derniere iteration
DEP = TTU . 'ESTIMATION' . 'DEPLACEMENTS' ;
REA = TTU . 'ESTIMATION' . 'REACTIONS' ;
WTAB = TTU . 'WTABLE' ;
** Calcul du jeu courant a la derniere iteration
JEU_I = ABS (JEU_INI + (EXTR DEP 'UX' P3) - (EXTR DEP 'UX' P2)) ;
** Calcul de la reaction au point P3
SI ((TTU . 'ESTIMATION' . 'TEMPS') NEG 0.) ;
FXL3 = (EXTR (RESU (REDU REA L3)) 'FX' P3) / EPAI ;
PARA_P = ABS (FXL3) ;
** On met a jour le chargement en PC
```



2.8. RÉSULTATS NUMÉRIQUES

```
CH_PC      = MANU 'CHPO' MRACC 'PC' PARA_P ;
CHA3       = CHAR 'PC' CH_PC EV1 ;
WTAB . 'CHARGEMENT' = (TTU . 'MES_DONNEES' . 'CHAR_MECA') ET CHA3 ;
FINS ;
FINP ;
```

On peut remarquer que la pression de contact utilisée ici est la force linéique sur la ligne L3.

Les valeurs du coefficient d'échange utilisées dans cet exemple ne sont pas nécessairement réalistes. On peut tout de même observer leur influence sur les résultats (cf. 2.8). L'objectif était simplement de montrer comment faire dépendre un paramètre matériau de variables représentant l'état mécanique du système. Si l'on souhaite faire dépendre un paramètre matériau de variables représentatives de la thermique, on pourra utiliser la procédure utilisateur REEV_THE qui est appelée après la résolution de la thermique et permet de ré-évaluer l'état du système selon l'état thermique calculé.

2.8 Résultats numériques

On observe l'évolution du jeu entre les barreaux au cours du temps ainsi que celle des températures aux extrémités des barreaux P2 et P3. La figure 2.2 présente l'évolution du jeu et des températures en P2 et P3 en fonction du temps obtenues par les deux approches de calcul : avec la procédure PERSO1 et la procédure REEV_MEC. Les valeurs du jeu et des températures obtenues avec REEV_MEC sont identiques à celles obtenues par PERSO1 mais avec un décalage temporel. En effet, chaque mise en contact ou ré-ouverture du jeu est estimée un pas de temps plus tard par la méthode PERSO1 par rapport à la méthode REEV_MEC.

La figure 2.3 présente les mêmes données obtenues en modélisant le transfert de chaleur entre les deux barreaux par convection. On peut observer que lors du contact, les températures de part et d'autre de l'interface diffèrent. Les durées des contacts sont plus faibles que dans le cas précédent.




FIG. 2.2 – Évolution du jeu entre les barreaux en fonction du temps (a) et des températures aux points P2 et P3 en fonction du temps (b). Le transfert de chaleur se fait par conduction lors du contact au moyen des procédures PERSO1 ou REEV_MEC.




FIG. 2.3 – Évolution du jeu entre les barreaux en fonction du temps (a) et des températures aux points P2 et P3 en fonction du temps (b). Le transfert de chaleur se fait par convection lors du contact au moyen de la procédure REEV_MEC.



2.9. JEU DE DONNÉES

```
CONF_0 = FORM ;
NBOU1 = DIME (TAB1 . 'TEMPS_CALCULES') ;
MESS ' Temps      Jeu      T2      T3' ;
** Listing du jeu et des temperatures pour les pas de temps
REPE BOU1 NBOU1 ;
I      = &BOU1 - 1 ;
TEMPS_I = TAB1 . 'TEMPS' . I ;
TEMP_I  = TAB1 . 'TEMPERATURES' . I ;
DEPL_I  = TAB1 . 'DEPLACEMENTS' . I ;
CONF_I  = FORM DEPL_I ;
JEU_I   = JEU_INI + (EXTR DEPL_I 'UX' P3) - (EXTR DEPL_I 'UX' P2) ;
T2      = EXTR TEMP_I 'T' P2 ;
T3      = EXTR TEMP_I 'T' P3 ;
MESS TEMPS_I JEU_I T2 T3 ;
SI GRAPH ;
LJEU    = LJEU ET JEU_I ;
LT2     = LT2 ET T2 ;
LT3     = LT3 ET T3 ;
** Trace du champ de temperature sur la configuration deformee
TRAC TEMP_I MAILT (PROG T4 PAS ((T1-T4)/20.) T1) 'TITR'

(CHAI 'Temperature au temps' TEMPS_I) ;
FINS ;
FORM CONF_0 ;
FIN BOU1 ;
SI GRAPH ;
** Temperature VS temps aux points P2 et P3
JEU_VST = EVOL 'MANU' 'Temps' (TAB1 . 'TEMPS_CALCULES') 'Jeu'
LJEU_VST = COUL 'ROUG' ;
T2_VST   = EVOL 'MANU' 'Temps' (TAB1 . 'TEMPS_CALCULES') 'Temperature'
LT2_VST  = COUL 'ROUG' ;
T3_VST   = EVOL 'MANU' 'Temps' (TAB1 . 'TEMPS_CALCULES') 'Temperature'
LT3_VST  = COUL 'BLEU' ;
TAB2     = TABL ;
TAB2 . 'TITRE' = TABL ;
TAB2 . 'TITRE' . 1 = 'Point P2' ;
TAB2 . 'TITRE' . 2 = 'Point P3' ;
DESS JEU_VST 'TITR' 'Jeu VS temps' ;
DESS (T2_VST ET T3_VST) 'TITR' 'Temperature VS temps' 'LEGE' TAB2 ;
FINS ;
FIN ;
```




Chapitre 3

Cas test 3 : Fissuration d'une poutre par un essai de flexion 4 points

3.1 Description

La figure 3.1 présente le problème de ce cas test. Un poutre de demi-longueur $L = 1$ m est sollicitée en flexion 4 points par des appuis cylindriques. Le déplacement des appuis supérieurs est imposé jusqu'à la fissuration de la poutre. Les fissures sont modélisées à l'aide d'une variable interne du modèle de plasticité qui représente la fissuration en traction. On cherchera à représenter l'ouverture de la fissure en supprimant les éléments finis, au cours du calcul, lorsque cette variable atteint une valeur seuil.




FIG. 3.1 – Illustration du problème.

3.2 Objectifs

- Tester la mécanique non linéaire avec un modèle de plasticité représentant la fissuration.
- Montrer comment réaliser un enlèvement de matière à l'aide de la procédure utilisateur PERSO1 dans PASAPAS.

3.3 Conditions initiales et aux limites

- Le problème est modélisé en 2D et en hypothèse de déformations planes.
- La moitié de la poutre est modélisée : conditions de symétrie $u_x = 0$ sur le côté gauche.
- Conditions de contact entre les appuis et la poutre.
- Déplacement vertical imposé de l'appuis supérieur $u_y = -0.05h$.
- Déplacement bloqué de l'appuis inférieur $u_x = u_y = 0$.

3.4 Propriétés mécaniques

Le modèle de plasticité DRUCKER_PRAGER_2 est utilisé pour représenter le comportement du béton. Il s'agit d'un modèle à deux surfaces de charges différentes selon que l'on est en traction ou en compression. Les propriétés mécaniques du béton, utilisées pour le calcul, sont présentées dans le tableau 3.1.

TAB. 3.1 – Propriétés mécaniques du béton.

| Propriété mécanique | Valeur |
|---|--------------------|
| Module de YOUNG (GPa) | 30 |
| Coefficient de POISSON | 0.2 |
| Limite d'élasticité (MPa) | 20 |
| Énergie de fissuration (J) | 100 |
| Résistance en traction simple (MPa) | 3.5 |
| Résistance en compression simple (MPa) | 35 |
| Résistance en compression bi-axiale (MPa) | 40.6 |
| Déformation au pic en compression | $-2 \cdot 10^{-3}$ |
| Déformation ultime en compression | $-8 \cdot 10^{-3}$ |

Un dernier paramètre du modèle, la longueur caractéristique des éléments finis, est calculé automatiquement selon le maillage (cf. 3.5.2).

3.5 Commentaires

3.5.1 Prise en compte du contact

La condition de contact entre les appuis et la poutre est imposée au moyen de l'opérateur IMPO :

```
MCONTH = IMPO 'MAIL' LHAUT APPUIH 'COUL' 'JAUN' ;
MCONTB = IMPO 'MAIL' LBAS APPUIB 'COUL' 'JAUN' ;
```

Les maillages ainsi créés doivent être passés en arguments de la table de PASAPAS à l'indice 'CONTACT' :

```
TAB1 . 'CONTACT' = MCONTH ET MCONTB ;
```

3.5.2 Calcul de la longueur caractéristique

Le modèle DRUCKER_PRAGER_2 fait intervenir un paramètre matériau 'LCAR' qui représente la longueur caractéristique des éléments finis du maillage utilisé. Ce paramètre peut être pris, en dimension 2, comme la racine carrée du jacobien des éléments. Le champ de longueur caractéristique est alors obtenu à l'aide de l'opérateur JACO, quel que soit le maillage utilisé :

```
** Calcul de la longueur caracteristique du maillage
JACO1 = JACO MODB ;
JACO2 = CHAN 'RIGIDITE' MODB JACO1 ;
LCAR = JACO2**(1./2.) ;
```

On remarquera que l'on prend soin d'exprimer le champ de longueur caractéristique aux points où sont calculées les rigidités des éléments à l'aide de l'opérateur CHANger.



3.5.3 Modélisation de la fissuration en traction

Le modèle DRUCKER_PRAGER_2 décrit la fissuration du matériau en traction à l'aide d'une variable interne nommée 'EPS2'. Il s'agit de la variable interne d'écroutissage en traction, notée χ_2 . À partir de cette variable interne, on peut calculer une autre variable χ , représentative de la fissuration en traction :

$$\chi = \exp\left(-\frac{\chi_2}{\chi_{2u}}\right) \quad (3.1)$$

où χ_{2u} est un paramètre du modèle dépendant de l'énergie de fissuration G_f , de la longueur caractéristique h et de la résistance en traction simple f_t :

$$\chi_{2u} = \frac{G_f}{h f_t} \frac{1}{1 - \exp(-1)} \quad (3.2)$$

La variable χ est comprise entre 0 et 1 et permet de quantifier la fissuration en traction dans le matériau. La valeur initiale $\chi = 1$ correspond au matériau sain, non endommagé. La valeur asymptotique $\chi = 0$ correspond à une fissuration totale du matériau en traction.

3.5.4 Élimination des éléments au cours du calcul (PERSO1)

Dans cet exemple, on montre comment supprimer, au cours du calcul, les éléments finis totalement fissurés. Pour cela, on peut utiliser la procédure utilisateur PERSO1 qui est exécutée à la fin du calcul de chaque pas de temps. Son utilisation est demandée par la commande :

```
TABL . 'PROCEDURE_PERSO1' = VRAI ;
```

Il s'agit alors de calculer, après chaque pas de temps, le champ de variable de fissuration χ et de supprimer les éléments pour lesquels ce champ est inférieur à une valeur seuil. Pour cela, on calcule, dans un premier temps, le champ de χ_{2u} sur le maillage et on le stocke dans la table de PASAPAS :

```
** Variable d'ecrouissage ultime en traction
X2U      = (LCAR**(-1))*GF/(LTR*(1.-(EXP (-1.)))) ;
X2U      = CHAN 'STRESSES' X2U MODB ;
...
TABL . 'MES_DONNEES'          = TABL ;
TABL . 'MES_DONNEES' . 'X2U' = X2U ;
```

On remarquera l'utilisation de l'opérateur CHANger pour exprimer le champ X2U aux points d'intégration des contraintes. En effet, on effectuera par la suite une opération de multiplication entre ce champ et le champ de variable interne χ_2 exprimé aux points d'intégration des contraintes : il est nécessaire que les deux champs soient exprimés aux mêmes points.

La procédure PERSO1 est ensuite définie pour modifier, comme on le souhaite, la table de PASAPAS :

```
DEBP PERSO1 TTU*'TABLE' ;
** Variables internes, modele et maillage a la derniere iteration
VI      = TTU . 'ESTIMATION' . 'VARIABLES_INTERNES' ;
MOD1    = TTU . 'WTABLE' . 'MOD_MEC' ;
MAIL1   = EXTR MOD1 'MAIL' ;
** Calcul du champ de fissuration en traction
X2      = CHAN 'TYPE' (EXCO VI 'EPS2') 'SCALAIRE' ;
X2U     = TTU . 'MES_DONNEES' . 'X2U' ;
X2U     = REDU X2U MOD1 ;
X       = EXP (-1.*X2*(X2U**(-1))) ;
```

```

** On determine les elements finis fissures
MAIL2 = X ELEM 'INFE' 1.5E-2 ;
** Calcul du nouveau maillage et reduction du modele sur ce dernier
NBE = NBEL MAIL2 ;
SI (NBE > 0) ;
MESS '***** Suppression de' NBE 'elements' ;
MAIL3 = DIFF MAIL1 MAIL2 ;
MOD2 = REDU MOD1 MAIL3 ;
** Mise a jour du modele mecanique dans la table de travail
TTU . 'WTABLE' . 'MOD_MEC' = MOD2 ;
TTU . 'WTABLE' . 'MO_TOT' = TTU . 'WTABLE' . 'MOD_MEC' ;
TTU . 'WTABLE' . 'MO_TOTAL' = TTU . 'WTABLE' . 'MOD_MEC' ;
FINS ;
FINP ;

```

Commentons cette procédure :

- Il s'agit d'abord de récupérer le champ de variable 'EPS2' à la dernière itération (contenu dans la table 'ESTIMATION' de la table de PASAPAS) ainsi que le modèle mécanique (contenu dans la table de travail 'WTABLE' à l'indice 'MOD_MEC').
- On calcule le champ de fissuration χ à partir du champ χ_2 (composante 'EPS2' des variables internes) et du champ de χ_{2ii} . On notera qu'il est nécessaire de changer le sous-type du champ de variables internes en sous-type 'SCALAIRE' afin de pouvoir effectuer l'opération de multiplication.
- On isole les éléments finis où le champ χ est inférieur à $1.5 \cdot 10^{-2}$ (opérateur ELEM option 'INFE'). Le nouveau maillage de la poutre est obtenu en calculant la différence entre le maillage initial et les éléments fissurés (opérateur DIFF).
- Il faut alors mettre à jour le modèle de la table de travail en réduisant le modèle mécanique sur ce nouveau maillage (opérateur REDU). Il faut également mettre à jours les indices 'MO_TOT' et 'MO_TOTAL' de la table de travail qui contiennent l'assemblage de tous les modèles. Il n'y a qu'un seul modèle dans ce cas test, mais il convient d'intégrer les autres modèles le cas échéant.

Notons que les champs de propriétés matérielles, ainsi que les champs de contraintes, de variables internes, etc., calculés aux itérations suivantes, seront automatiquement réduits sur le nouveau modèle, c'est pourquoi nous ne faisons pas cette opération dans la procédure PERSO1.

Précisons également que le choix de la valeur seuil pour supprimer les éléments finis fissurés peut s'avérer délicat. Un critère moins sévère, ou un chargement plus important, pourra conduire à une fissure traversante, c'est-à-dire créant une séparation de la poutre en deux solides, ce qui posera problème lors du calcul, le milieu étant supposé continu.

3.6 Résultats numériques

La figure 3.2 présente les courbes force-déplacement de l'essai de flexion obtenues en utilisant la procédure PERSO1 (élimination des éléments finis fissurés) et en utilisant l'approche classique. Les deux approches donnent des résultats similaires. La figure 3.3 présente les champs de la variable de fissuration en traction χ obtenus par les deux approches à la fin du calcul.




FIG. 3.2 – Courbes force-déplacement de l'essai de flexion obtenues selon l'approche avec élimination des éléments fissurés par la procédure PERSO1 et selon l'approche classique.




FIG. 3.3 – Champ de fissuration en traction χ pour une flèche de 6 mm selon l'approche avec élimination des éléments fissurés par la procédure PERSO1 (a) et selon l'approche classique (b). La déformée est amplifiée 10 fois.

3.7 Jeu de données

```

OPTI 'ECHO' 0 ;
*****
*
*          Essai de flexion 4 points sur une poutre en beton
*
*****
* Calcul mecanique non lineaire (endommagement).
* Utilisation de la procedure PASAPAS.
* Utilisation de la procedure utilisateur PERSOL.
*
*
* Caracteristiques de la poutre (beton, modele Drucker Prager 2) :
*
* Longueur                L      = 1. m
* Epaisseur               h      = 0.1 m
* Module de Young         E      = 30 GPa
* Coefficient de Poisson  Nu     = 0.2
* Limite d'elasticite     Sigy  = 20 MPa
* Energie de fissuration  Gf    = 100. N.m
* Resistance en traction simple  LTR  = 3.5 MPa
* Resistance en compression simple  LCS  = 35 MPa
* Resistance en compression bi-axiale  LBI  = 1.16 * LCS
* Deformation au pic en compression simple  EPM  = -2.E-3
* Deformation ultime en compression simple  EPU  = -8.E-3
*
* Le parametre LCAR, representant la longueur caracteristique des
* elements finis, est calcule automatiquement en fonction du maillage.
*
*
*----- Options generales -----*
*
OPTI 'DIME' 2 'ELEM' 'QUA4' 'MODE' 'PLAN' 'DEFO' ;
** Indicateur de trace
GRAPH = VRAI ;
*
*----- Parametres geometriques, materiaux et CL -----*
*
** Geometrie
L      = 1. ;
H      = 0.1 ;
RAPP  = 0.05 ;
** Materiau
YOUN  = 30.E9 ;
NU    = 0.2 ;
SIGY  = 20.E6 ;
LTR   = 3.5E6 ;
GF    = 100. ;
LCS   = 3.5E7 ;
LBI   = 1.16*LCS ;
EPM   = -2.E-3 ;
EPU   = -8.E-3 ;
** Conditions aux limites
UY_IMP = -1.*0.05*H ;
*
*----- Maillage -----*
*
** Beton
P0    = 0. 0. ;
P1    = (0.2*L) 0. ;
P2    = (0.8*L) 0. ;
P3    = L 0. ;
DENS1 = L/100. ;
DENS2 = L/200. ;
DENS3 = L/10. ;
L1    = DROI P0 'DINI' DENS1 P1 'DFIN' DENS2 ;
L2    = DROI L1 'DINI' DENS2 P2 'DFIN' DENS3 ;
LBAS  = DROI L2 'DINI' DENS3 P3 'DFIN' DENS3 ;
MAILB = (LBAS TRAN 11 (0. H) COUL 'BLEU' ;
LHAUT = MAILB COTE 3 ;
LGAUCHE = MAILB COTE 4 ;
** Appuis
PAPPH = P1 PLUS (0. (H+RAPP)) ;
APPUIH = CERC 10 (PAPPH MOIN (0. RAPP)) PAPPH (PAPPH PLUS (RAPP 0.)) ;
APPUIH = APPUIH ET (APPUIH TOUR PAPPH -90.) ;
ELIM 1.E-9 APPUIH ;
PAPPB = P2 PLUS (0. (-1.*RAPP)) ;
APPUIB = APPUIH TOUR PAPPH 180. ;
DEPL APPUIB 'PLUS' (PAPPB MOIN PAPPH) ;
** Contact
MCONTH = IMPO 'MAIL' LHAUT APPUIH 'COUL' 'JAUN' ;
MCONTB = IMPO 'MAIL' LBAS APPUIB 'COUL' 'JAUN' ;
SI GRAPH ;
NBE    = NBEL MAILB ;
TRAC (MAILB ET MCONTH ET MCONTB) 'TITR'
      (CHAI 'Maillage (' NBE ' elements)') ;
FINS ;
*
*----- Modeles et materiaux -----*
*
MODB   = MODE MAILB 'MECANIQUE' 'ELASTIQUE' 'ISOTROPE' 'PLASTIQUE'
        'DRUCKER_PRAGER_2' ;
** Calcul de la longueur caracteristique du maillage
JACO1  = JACO MODB ;
JACO2  = CHAN 'RIGIDITE' MODB JACO1 ;
LCAR   = JACO2**(1./2.) ;
MATB   = MATE MODB 'YOUN' 'NU' 'NU' 'GF' 'GF' 'SIGY' 'SIGY' 'LTR' 'LTR'
        'LCS' 'LCS' 'LBI' 'LBI' 'EPM' 'EPM' 'EPU' 'EPU' 'LCAR' 'LCAR' ;
** Variable d'ecrouissage ultime en traction
X2U    = (LCAR**(-1))*GF/(LTR*(1.-(EXP (-1.)))) ;
X2U    = CHAN 'STRESSES' X2U MODB ;
*
*----- Blocages mecaniques -----*
*
CL_SYM = BLOQ (APPUIH ET LGAUCHE) 'UX' ;
CL_APPH = BLOQ APPUIH 'UY' ;
CL_APPB = BLOQ APPUIB 'DEPL' ;
DEPIH  = DEPI CL_APPH UY_IMP ;
SI GRAPH ;
VECL1  = VECT (DEPIH NOMC 'UY') 5. 'UX' 'UY' 'VERT' ;
TRAC VECL1 ((CONT (MAILB)) ET APPUIH ET APPUIB)
           'TITR' 'Deplacement impose (x5)' ;
FINS ;
*
*----- Chargement en deplacements imposees -----*
*
LTPS   = PROG 0. PAS 0.05 0.2 PAS 0.02 0.8 PAS 0.05 1. ;
LAMP   = LTPS ;
EV1    = EVOL 'MANU' 'TEMPS' LTPS 'AMPL' LAMP ;
CHAR1  = CHAR 'DIMP' DEPIH EV1 ;
*
*----- Creation de la table pour PASAPAS -----*
*
TAB1   =
TAB1 . 'MODELE'           = TABL ;
TAB1 . 'CARACTERISTIQUES' = MODB ;
TAB1 . 'CARACTERISTIQUES' = MATB ;
TAB1 . 'BLOCAGES_MECAIQUES' = CL_SYM ET CL_APPH ET CL_APPB ;
TAB1 . 'CHARGEMENT'       = CHAR1 ;
TAB1 . 'CONTACT'         = MCONTH ET MCONTB ;
TAB1 . 'PROCEDURE_PERSOL' = VRAI ;
TAB1 . 'TEMPS_CALCULES'   = LTPS ;
TAB1 . 'MES_DONNEES'     = TABL ;
TAB1 . 'MES_DONNEES' . 'X2U' = X2U ;
*
*----- Procedure PERSOL -----*
*
DEBP PERSOL TTU*'TABLE' ;
** Variables internes, modele et maillage a la derniere iteration
VI     = TTU . 'ESTIMATION' . 'VARIABLES_INTERNES' ;
MOD1   = TTU . 'WTABLE' . 'MOD_MEC' ;
MAIL1  = EXTR MOD1 'MAIL' ;
** Calcul du champ de fissuration en traction
X2     = CHAN 'TYPE' (EXCO VI 'EPS2') 'SCALAIRE' ;
X2U    = TTU . 'MES_DONNEES' . 'X2U' ;
X2U    = REDU X2U MOD1 ;
X      = EXP (-1.*X2*(X2U**(-1))) ;
** On determine les elements finis fissures
MAIL2  = X ELEM 'INFE' 1.5E-2 ;
** Calcul du nouveau maillage et reduction du modele sur ce dernier
NBE    = NBEL MAIL2 ;
SI (NBE > 0) ;
MESS '***** Suppression de' NBE 'elements' ;
MAIL3  = DIFF MAIL1 MAIL2 ;
MOD2   = REDU MOD1 MAIL3 ;
** Mise a jour du modele mecanique dans la table de travail
TTU . 'WTABLE' . 'MOD_MEC' = MOD2 ;
TTU . 'WTABLE' . 'MO_TOT'  = TTU . 'WTABLE' . 'MOD_MEC' ;
TTU . 'WTABLE' . 'MO_TOTAL' = TTU . 'WTABLE' . 'MOD_MEC' ;
FINS ;
FINP ;
*
*----- Appel à la procedure PASAPAS -----*
*
PASAPAS TAB1 ;
*
*----- Post-traitement -----*
*
CONF_0 = FORM ;
LFLECHE = PROG 0. ;
LFORCE = PROG 0. ;
** On calcule la fleche, la force de reaction et le champ de fissuration
** en traction pour chaque pas de temps
REPE BOUL ((DIME (TAB1 . 'TEMPS')) - 1) ;
I      = &BOUL ;
TPS_I  = TAB1 . 'TEMPS' . I ;
DEPL_I = TAB1 . 'DEPLACEMENTS' . I ;
REAC_I = TAB1 . 'REACTIONS' . I ;
VI_I   = TAB1 . 'VARIABLES_INTERNES' . I ;
X2     = CHAN 'TYPE' (EXCO VI_I 'EPS2') 'SCALAIRE' ;
MAIL1  = EXTR X2 'MAIL' ;
MOD1   = REDU MODB MAIL1 ;
X2U1   = REDU X2U MOD1 ;
X      = EXP (-1.*X2*(X2U1**(-1))) ;
LFLECHE = LFLECHE ET (-1.*(EXTR DEPL_I 'UY' P0)) ;
LFORCE = LFORCE ET (2.*(EXTR REAC_I 'FY' P2)) ;
CONF_I = FORM (10.*DEPL_I) ;
SI GRAPH ;

```



3.7. JEU DE DONNÉES

```
** Trace du champ de fissuration sur la configuration deformee (x10)
** Ce champ prend des valeurs comprises entre 0 et 1
** 1 -> pas de fissuration en traction, materiau sain
** 0 -> materiau totalement fissure en traction
    TRAC X MOD1 (PROG 0. PAS 0.05 1.)
    'TITR' (CHAI 'Fissuration en traction au temps' TPS_I) ;
FINS ;

    FORM CONF_0 ;
    FIN BOU1 ;
    FVSU      = EVOL 'MANU' 'Fleche' LFLECHE 'Force' LFORCE ;
    DESS FVSU 'TITR' 'Courbe force - Deplacement' ;
    FIN ;
```

